ROCKET

LAND AN APOLLO CAPSULE ON THE MOON

Description

ROCKET, known also as LUNAR, LEM, and APOLLO, is by far and away the single most popular computer game. It exists in versions that start you anywhere from 500 feet to 200 miles above the moon, or other planets, too. Some allow the control of directional stabilization rockets and/or the retro rocket. The three versions presented here appear to be the most popular of the many variations.

ROCKET. In this program, you set the burn rate of the retro rockets (pounds of fuel per second) every 10 seconds and attempt to achieve a soft landing on the moon. 200 lbs/sec really puts the brakes on, and 0 lbs/sec is free fall. Ignition occurs at 8 lbs/sec, so do not use burn rates between 1 and 7 lbs/sec. To make the landing more of a challenge, but more closely approximate the real Apollo LEM capsule, you should make the available fuel at the start (N) equal to 16,000 lbs, and the weight of the capsule (M) equal to 32,500 lbs in Statement 15.

Some computers object to the series expansion calculations in Statements 91 and 94 (as you near the lunar surface, these numbers get very small). If yours does, substitute the expanded form--for the expansion in Statement 91:

```
-Q*(1+Q*(1/2+Q*(1/3+Q*(1/4+Q/5))))
```

You should be able to figure the other one out yourself.

ROCKT1. In this version, you start 500 feet above the lunar surface and control the burn rate in 1-second bursts. Each unit of fuel slows your descent by 1 ft/sec. The maximum thrust of your engine is 30 ft/sec/sec.

ROCKT2. This is the most comprehensive of the three versions and permits you to control the time interval of firing, the thrust, and the attitude angle. It also allows you to work in the metric or English system of measurement. The instructions in the program dialog are very complete, so you shouldn't have any trouble.

In most versions of ROCKET, the temptation is to slow up too soon and then have no fuel left for the lower part of the journey. This, of course, is disasterous (as you will find out when you land your own capsule)!

Source

To put all the conflicting stories to rest, we can say with confidence that ROCKET was originally written in FOCAL by a Lexington High School student back in the mid 60's.

ROCKET:
Jim Storer
Lexington High School
Lexington, MA 02173

ROCKT1:
Eric Peters
Digital Equipment Corp.
Maynard, MA 01754

ROCKT2:
William Labaree II
621 Oakley Place
Alexandria, VA 22302



ROCKET PROGRAM LISTING

```
ROCKET  EDUSYSTEM 30

2 PRINT "THIS IS A COMPUTER SIMULATION OF AN APOLLO LUNAR"
3 PRINT "LANDING CAPSULE."\PRINT\PRINT
4 PRINT "THE ON-BOARD COMPUTER HAS FAILED (IT WASN'T MADE BY"
5 PRINT "DIGITAL) SO YOU HAVE TO LAND THE CAPSULE MANUALLY"
6 PRINT\PRINT "SET BURN RATE OF RETRO ROCKETS TO ANY VALUE BETWEEN"
7 PRINT "0 (FREE FALL) AND 200 (MAXIMUM BURN) POUNDS PER SECOND"
8 PRINT "SET NEW BURN RATE EVERY 10 SECONDS."\PRINT
9 PRINT "CAPSULE WEIGHT 32,500 LBS;  FUEL WEIGHT 16,500 LBS"
10 PRINT\PRINT\PRINT "GOOD LUCK!!!"
11 L=0
13 PRINT\PRINT "SEC","MI + FT","MPH","LB FUEL","BURN RATE"\PRINT
15 A=120\V=1\M=33000\N=16500\G=1E-3\Z=1.8
21 PRINT L,INT(A);INT(5280*(A-INT(A))),3600*V,M-N,\INPUT K\T=10
31 IF M-N<.001 THEN 41\IF T<.001 THEN 21\S=T\IF M>=N+S*K THEN 35
32 S=(M-N)/K
35 GOSUB 91\IF I<=0 THEN 71\IF V<=0 THEN 38\IF J<0 THEN 81
38 GOSUB 61\GOTO 31
41 PRINT "FUEL OUT AT"L"SEC"\S=(-V+SQR(V*V+2*A*G))/G\V=V+G*S\L=L+S
51 W=3600*V\PRINT"ON MOON AT"L"SEC - IMPACT VELOCITY" W "MPH"
52 IF W>1.2 THEN 53\PRINT "PERFECT LANDING! (LUCKY)"\GOTO 95
53 IF W>10 THEN 56\PRINT "GOOD LANDING (COULD BE BETTER)"\GOTO 95
56 IF W>60 THEN 58 \PRINT "CRAFT DAMAGE. . . . . YOU'RE STRANDED HERE UNTIL"
57 PRINT "A RESCUE PARTY ARRIVES.  HOPE YOU HAVE ENOUGH OXYGEN!"\GOTO 95
58 PRINT "SORRY, BUT THERE WERE NO SURVIVORS. . . YOU BLEW IT!"
59 PRINT "IN FACT, YOU BLASTED A NEW LUNAR CRATER"W*.2777"FT DEEP"
60 GOTO 95
61 L=L+S\T=T-S\M=M-S*K\A=I\V=J\RETURN
71 IF S<5E-3 THEN 51\D=V+SQR(V*V+2*A*(G-Z*K/M))\S=2*A/D
73 GOSUB 91\GOSUB 61\GOTO 71
81 W=(1-M*G/(Z*K))/2\S=M*V/(Z*K*(W+SQR(W*W+V/Z)))+.05\GOSUB 91
83 IF I<=0 THEN 71\GOSUB 61\IF J>0 THEN 31\IF V>0 THEN 81\GOTO 31
91 Q=S*K/M\J=V+G*S+Z*(-Q-Q*Q/2-Q^3/3-Q^4/4-Q^5/5)
94 I=A-G*S*S/2-V*S+Z*S*(Q/2+Q^2/6+Q^3/12+Q^4/20+Q^5/30)\RETURN
95 PRINT\PRINT\PRINT\PRINT "TRY AGAIN??"\GOTO 6
99 END
```

SAMPLE RUN

```
ROCKET  EDUSYSTEM 30

THIS IS A COMPUTER SIMULATION OF AN APOLLO LUNAR
LANDING CAPSULE.


THE ON-BOARD COMPUTER HAS FAILED (IT WASN'T MADE BY
DIGITAL) SO YOU HAVE TO LAND THE CAPSULE MANUALLY

SET BURN RATE OF RETRO ROCKETS TO ANY VALUE BETWEEN
0 (FREE FALL) AND 200 (MAXIMUM BURN) POUNDS PER SECOND
SET NEW BURN RATE EVERY 10 SECONDS.

CAPSULE WEIGHT 32,500 LBS;  FUEL WEIGHT 16,500 LBS


GOOD LUCK!!!

SEC           MI + FT       MPH           LB FUEL       BURN RATE

 0             120  0        3600          16500         ?0
 10            109  5015     3636          16500         ?0
 20            99  4223      3672          16500         ?0
 30            89  2903      3708          16500         ?0
 40            79  1055      3744          16500         ?0
 50            68  3959      3780          16500         ?0
 60            58  1055      3816          16500         ?0
 70            47  2903      3852          16500         ?200
 80            37  1883      3482.87       14500         ?200
 90            28  1191      3086.7        12500         ?200
 100           20  1251      2659.65       10500         ?200
 110           13  2549      2196.94       8500          ?200
 120           8  370        1692.63       6500          ?200
 130           4  658        1139.13       4500          ?200
 140           1  4203       526.598       2500          ?100
 150           0  4042       212.242       1500          ?45
 160           0  1863       84.1831       1050          ?20
 170           0  908        45.9129       850           ?17
 180           0  438        18.107        680           ?12
 190           0  241        8.68632       560           ?11
 200           0  157        2.7691        450           ?9
 210           0  105        4.27036       360           ?9.5
 220           0  46         3.65466       265           ?9.8
 230           0  7          1.66462       167           ?9.3
ON MOON AT 233.183 SEC - IMPACT VELOCITY 1.6042 MPH
GOOD LANDING (COULD BE BETTER)


TRY AGAIN??
```

ROCKT1 PROGRAM LISTING

```
LIST
ROCKT1  03:37 PM        08-MAY-73
70 PRINT "LUNAR LANDING SIMULATION"
80 PRINT "----- ------- ----------"\PRINT
100 INPUT "DO YOU WANT INSTRUCTIONS (YES OR NO)";A$
120 IF A$="NO" THEN 390
160 PRINT
200 PRINT "YOU ARE LANDING ON THE MOON AND HAVE TAKEN OVER MANUAL"
210 PRINT "CONTROL 500 FEET ABOVE A GOOD LANDING SPOT.  YOU HAVE A"
220 PRINT "DOWNWARD VELOCITY OF 50 FT/SEC.  120 UNITS OF FUEL REMAIN."
225 PRINT
230 PRINT "HERE ARE THE RULES THAT GOVERN YOUR SPACE VEHICLE:"
240 PRINT "(1) AFTER EACH SECOND, THE HEIGHT, VELOCITY, AND REMAINING"
250 PRINT "    FUEL WILL BE REPORTED."
260 PRINT "(2) AFTER THE REPORT, A '?' WILL BE TYPED.  ENTER THE"
270 PRINT "    NUMBER OF UNITS OF FUEL YOU WISH TO BURN DURING THE"
280 PRINT "    NEXT SECOND.  EACH UNIT OF FUEL WILL SLOW YOUR DESCENT"
290 PRINT "    BY 1 FT/SEC."
310 PRINT "(3) THE MAXIMUM THRUST OF YOUR ENGINE IS 30 FT/SEC/SEC OR"
320 PRINT "    30 UNITS OF FUEL PER SECOND."
330 PRINT "(4) WHEN YOU CONTACT THE LUNAR SURFACE, YOUR DESCENT ENGINE"
340 PRINT "    WILL AUTOMATICALLY CUT OFF AND YOU WILL BE GIVEN A"
350 PRINT "    REPORT OF YOUR LANDING SPEED AND REMAINING FUEL."
360 PRINT "(5) IF YOU RUN OUT OF FUEL, THE '?' WILL NO LONGER APPEAR,"
370 PRINT "    BUT YOU SECOND BY SECOND REPORT WILL CONTINUE UNTIL"
380 PRINT "    YOU CONTACT THE LUNAR SURFACE."\PRINT
390 PRINT "BEGINNING LANDING PROCEDURE. . . . ."\PRINT
410 PRINT "G O O D   L U C K ! ! !"
420 PRINT\PRINT
440 PRINT "SEC  FEET   SPEED    FUEL      PLOT OF DISTANCE"
450 PRINT
455 T=0\H=500\V=50\F=120
490 PRINT T;TAB(4);H;TAB(12);V;TAB(20);F;TAB(29);"I";TAB(H/12+29);"*"
500 INPUT B
510 IF B<0 THEN 650
520 IF B>30 THEN B=30
530 IF B>F THEN B=F
540 V1=V-B+5
560 F=F-B
570 H=H-.5*(V+V1)
580 IF H<=0 THEN 670
590 T=T+1
600 V=V1
610 IF F>0 THEN 490
615 IF B=0 THEN 640
620 PRINT "*** OUT OF FUEL ***"
640 PRINT T;TAB(4);H;TAB(12);V;TAB(20);F;TAB(29);"I";TAB(H/12+29);"*"
650 B=0
660 GOTO 540
670 PRINT "*** CONTACT ***"
680 H=H+.5*(V+V1)
690 IF B=5 THEN 720
700 D=(-V+SQR(V*V+H*(10-2*B)))/(5-B)
710 GOTO 730
720 D=H/V
730 V1=V+(5-B)*D
760 PRINT "TOUCHDOWN AT";T+D;"SECONDS."
770 PRINT "LANDING VELOCITY =";V1;"FT/SEC"
780 PRINT F;"UNITS OF FUEL REMAINING."
790 IF V1<>0 THEN 810
800 PRINT "CONGRATULATIONS!!    A PERFECT LANDING!"
805 PRINT "YOUR LICENSE WILL BE RENEWED. . . . . LATER."
810 IF ABS(V1)<2 THEN 840
820 PRINT "***** SORRY, BUT YOU BLEW IT!!!!"
830 PRINT "APPROPRIATE CONDOLENCES WILL BE SENT TO YOUR NEXT OF KIN."
840 PRINT\PRINT\PRINT
850 INPUT "ANOTHER MISSION";A$
870 IF A$="YES" THEN 390
880 PRINT\PRINT "CONTROL OUT."\PRINT
999 END

READY
```

SAMPLE RUN

```
RUN
ROCKT1  03:39 PM        08-MAY-73
LUNAR LANDING SIMULATION
----- ------- ----------

DO YOU WANT INSTRUCTIONS (YES OR NO)? YES

YOU ARE LANDING ON THE MOON AND HAVE TAKEN OVER MANUAL
CONTROL 500 FEET ABOVE A GOOD LANDING SPOT.  YOU HAVE A
DOWNWARD VELOCITY OF 50 FT/SEC.  120 UNITS OF FUEL REMAIN.

HERE ARE THE RULES THAT GOVERN YOUR SPACE VEHICLE:
(1) AFTER EACH SECOND, THE HEIGHT, VELOCITY, AND REMAINING
    FUEL WILL BE REPORTED.
(2) AFTER THE REPORT, A '?' WILL BE TYPED.  ENTER THE
    NUMBER OF UNITS OF FUEL YOU WISH TO BURN DURING THE
    NEXT SECOND.  EACH UNIT OF FUEL WILL SLOW YOUR DESCENT
    BY 1 FT/SEC.
(3) THE MAXIMUM THRUST OF YOUR ENGINE IS 30 FT/SEC/SEC OR
    30 UNITS OF FUEL PER SECOND.
(4) WHEN YOU CONTACT THE LUNAR SURFACE, YOUR DESCENT ENGINE
    WILL AUTOMATICALLY CUT OFF AND YOU WILL BE GIVEN A
    REPORT OF YOUR LANDING SPEED AND REMAINING FUEL.
(5) IF YOU RUN OUT OF FUEL, THE '?' WILL NO LONGER APPEAR,
    BUT YOUR SECOND BY SECOND REPORT WILL CONTINUE UNTIL
    YOU CONTACT THE LUNAR SURFACE.

BEGINNING LANDING PROCEDURE. . . . .

G O O D   L U C K ! ! !

SEC  FEET   SPEED    FUEL      PLOT OF DISTANCE

 0   500     50      120     I                                          *
? 3
 1   449     52      117     I                                       *
? 3
 2   396     54      114     I                                    *
? 3
 3   341     56      111     I                                *
? 3
 4   284     58      108     I                           *
? 7
 5   227     56      101     I                      *
? 9
 6   173     52      92      I                 *
? 9
 7   123     48      83      I           *
? 8
 8   76.5    45      75      I      *
? 25
 9   41.5    25      50      I  *
? 25
 10  26.5    5       25      I *
? 25
*** OUT OF FUEL ***
 11  31.5   -15      0       I *
 12  44     -10      0       I  *
 13  51.5   -5       0       I   *
 14  54      0       0       I   *
 15  51.5    5       0       I   *
 16  44      10      0       I  *
 17  31.5    15      0       I *
 18  14      20      0       I*
*** CONTACT ***
TOUCHDOWN AT 18.6476 SECONDS.
LANDING VELOCITY = 23.2379 FT/SEC
 0 UNITS OF FUEL REMAINING.
***** SORRY, BUT YOU BLEW IT!!!!
APPROPRIATE CONDOLENCES WILL BE SENT TO YOUR NEXT OF KIN.


ANOTHER MISSION? YES
BEGINNING LANDING PROCEDURE. . . . .

G O O D   L U C K ! ! !

SEC  FEET   SPEED    FUEL      PLOT OF DISTANCE

 0   500     50      120     I                                          *
? 5
 1   450     50      115     I                                       *
? 5
 2   400     50      110     I                                    *
? 5
 3   350     50      105     I                                *
? 5
 4   300     50      100     I                           *
? 5
 5   250     50      95      I                      *
? 5
 6   200     50      90      I                 *
? 5
 7   150     50      85      I            *
? 5
 8   100     50      80      I       *
? 5
 9   50      50      75      I   *
? 30
 10  12.5    25      45      I*
? 30
*** CONTACT ***
TOUCHDOWN AT 11 SECONDS.
LANDING VELOCITY = 0 FT/SEC
 15 UNITS OF FUEL REMAINING.
CONGRATULATIONS!!    A PERFECT LANDING!
YOUR LICENSE WILL BE RENEWED. . . . . LATER.
```

ROCKT2 PROGRAM LISTING

```
 7 REM LUNARI IS A INTERACTIVE GAME THAT SIMULATES A LUNAR
 8 REM LANDING SIMILAR TO THAT OF THE APOLLO PROGRAM.
 9 REM THERE IS ABSOLUTELY NO CHANCE INVOLVED.
10 LET Z$="GO"
15 LET B1=1
20 LET M=17.95
25 LET F1=5.25
30 LET N=7.5
35 LET R0=926
40 LET V0=1.29
45 LET T=0
50 LET H0=60
55 LET R=R0+H0
60 LET A=-3.425
65 LET R1=0
70 LET A1=8.84361E-04
75 LET R3=0
80 LET A3=0
85 LET M1=7.45
90 LET M0=M1
95 LET B=750
100 LET T1=0
105 LET F=0
110 LET P=0
115 LET N=1
120 LET M2=0
125 LET S=0
130 LET C=0
135 IF Z$="YES" THEN 1150
140 PRINT
145 PRINT "LUNAR LANDING SIMULATION"
150 PRINT
155 PRINT "HAVE YOU FLOWN ON AN APOLLO/LEM MISSION BEFORE#";
160 PRINT "(YES OR NO)";
165 INPUT Q$
170 IF Q$="YES" THEN 190
175 IF Q$="NO" THEN 205
180 PRINT "JUST ANSWER THE QUESTION, PLEASE";
185 GOTO 160
190 PRINT
195 PRINT "ENTER MEASUREMENT OPTION NUMBER";
200 GOTO 225
205 PRINT
210 PRINT "WHICH SYSTEM OF MEASUREMENT DO YOU PREFER ?"
215 PRINT " 1=METRIC     0=ENGLISH"
220 PRINT "ENTER THE APPROPRIATE NUMBER";
225 INPUT K
230 PRINT
235 IF K=0 THEN 280
240 IF K=1 THEN 250
245 GOTO 220
250 LET Z=1852.8
255 LET M$="METERS"
260 LET G3=3.6
265 LET N$=" KILOMETERS"
270 LET G5=1000
275 GOTO 305
280 LET Z=6080
285 LET M$="FEET"
290 LET G3=.592
295 LET N$=" N.MILES"
300 LET G5=Z
305 IF B1=3 THEN 670
310 IF Q$="YES" THEN 485
315 PRINT
320 PRINT " YOU ARE ON A LUNAR LANDING MISSION.  AS THE PILOT OF"
325 PRINT "THE LUNAR EXCURSION MODULE, YOU WILL BE EXPECTED TO"
330 PRINT "GIVE CERTAIN COMMANDS TO THE MODULE NAVIGATION SYSTEM."
335 PRINT " THE ON BOARD COMPUTER WILL GIVE A RUNNING ACCOUNT"
340 PRINT "OF INFORMATION NEEDED TO NAVIGATE THE SHIP."
345 PRINT
350 PRINT
355 PRINT "THE ATTITUDE ANGLE CALLED FOR IS DESCRIBED AS FOLLOWS-"
360 PRINT "+ OR -180 DEGREES IS DIRECTLY AWAY FROM THE MOON"
365 PRINT "-90 DEGREES IS ON A TANGENT IN THE DIRECTION OF ORBIT"
370 PRINT "90 DEGREES IS ON A TANGENT FROM THE DIRECTION OF ORBIT"
375 PRINT "0 (ZERO) DEGREES IS DIRECTLY TOWARD THE MOON"
380 PRINT
385 PRINT TAB(30);"-180,180"
390 PRINT TAB(34);"↑"
395 PRINT TAB(27);"-90 < -+- > 90"
400 PRINT TAB(34);"!"
405 PRINT TAB(34);"0"
410 PRINT TAB(23);"<< DIRECTION OF ORBIT <<"
415 PRINT
420 PRINT TAB(27);"SURFACE OF MOON"
425 PRINT
430 PRINT
435 PRINT "ALL ANGLES BETWEEN -180 AND 180 DEGREES ARE ACCEPTED."
440 PRINT
445 PRINT "1 FUEL UNIT = 1 SEC. AT MAX. THRUST"
450 PRINT "ANY DISCREPANCIES ARE ACCOUNTED FOR IN THE USE OF FUEL"
455 PRINT "FOR AN ATTITUDE CHANGE."
460 PRINT "AVAILABLE ENGINE POWER: 0 (ZERO) AND ANY VALUE BETWEEN"
465 PRINT "10 AND 100 PERCENT"
470 PRINT
475 PRINT "NEGATIVE THRUST OR TIME IS PROHIBITED"
480 PRINT
485 PRINT
490 PRINT "INPUT: TIME INTERVAL IN SECONDS ------ (T)"
495 PRINT "       PERCENTAGE OF THRUST ---------- (P)"
500 PRINT "       ATTITUDE ANGLE IN DEGREES ----- (A)"
505 PRINT
510 IF Q$="YES" THEN 535
515 PRINT "FOR EXAMPLE:"
520 PRINT "T,P,A?10,65,-60"
525 PRINT "TO ABORT THE MISSION AT ANY TIME, ENTER 0,0,0"
530 PRINT
535 PRINT "OUTPUT: TOTAL TIME ELAPSED IN SECONDS"
540 PRINT "        HEIGHT IN ";M$
545 PRINT "        DISTANCE FROM LANDING SITE IN ";M$
550 PRINT "        VERTICAL VELOCITY IN ";M$;"/SECOND"
555 PRINT "        HORIZONTAL VELOCITY IN ";M$;"/SECOND"
560 PRINT "        FUEL UNITS REMAINING"
565 PRINT
570 GOTO 670
575 PRINT
580 PRINT "T,P,A";
585 INPUT T1,F,P
590 LET F=F/100
595 IF T1<0 THEN 905
600 IF T1=0 THEN 1090
605 IF ABS(F-.05)>1 THEN 945
610 IF ABS(F-.05)<.05 THEN 945
615 IF ABS(P)>180 THEN 925
620 LET N=20
625 IF T1<400 THEN 635
630 LET N=T1/20
635 LET T1=T1/N
640 LET P=P*3.14159/180
645 LET S=SIN(P)
650 LET C=COS(P)
655 LET M2=M0*T1*F/B
660 LET R3=-.5*R0*((V0/R)↑2)+R*A1*A1
665 LET A3=-2*R1*A1/R
670 FOR I=1 TO N
675 IF M1=0 THEN 715
680 LET M1=M1-M2
685 IF M1>0 THEN 725
690 LET F=F*(1+M1/M2)
695 LET M2=M1+M2
700 PRINT "YOU ARE OUT OF FUEL"
705 LET M1=0
710 GOTO 725
715 LET F=0
720 LET M2=0
725 LET M=M-.5*M2
730 LET R4=R3
735 LET R3=-.5*R0*((V0/R)↑2)+R*A1*A1
740 LET R2=(3*R3-R4)/2+.00526*F1*F*C/M
745 LET A4=A3
750 LET A3=-2*R1*A1/R
755 LET A2=(3*A3-A4)/2+.00526*F1*F*S/(M*R)
760 LET X=R1*T1+.5*R2*T1*T1
765 LET R=R+X
770 LET H0=H0+X
775 LET R1=R1+R2*T1
780 LET A=A+A1*T1+.5*A2*T1*T1
785 LET A1=A1+A2*T1
790 LET M=M-.5*M2
795 LET T=T+T1
800 IF H0<3.287828E-04 THEN 810
805 NEXT I
810 LET H=H0*Z
815 LET H1=R1*Z
820 LET D=R0*A*Z
825 LET D1=R*A1*Z
830 LET T2=M1*B/M0
835 PRINT TAB(1);T;TAB(10);H;TAB(23);D;
840 PRINT TAB(37);H1;TAB(49);D1;TAB(60);T2
845 IF H0<3.287828E-04 THEN 880
850 IF R0*A>164.4736 THEN 1050
855 IF M1>0 THEN 580
860 LET T1=20
865 LET F=0
870 LET P=0
875 GOTO 620
880 IF R1<-8.21957E-04 THEN 1020
885 IF ABS(R*A1)>4.931742E-04 THEN 1020
890 IF H0<-3.287828E-04 THEN 1020
895 IF ABS(D)>10*Z THEN 1065
900 GOTO 995
905 PRINT
910 PRINT "THIS SPACECRAFT IS NOT ABLE TO VIOLATE THE SPACE-";
915 PRINT "TIME CONTINUUM"
920 GOTO 575
925 PRINT
930 PRINT "IF YOU WANT TO SPIN AROUND, GO OUTSIDE THE MODULE";
935 PRINT "FOR AN E.V.A"
940 GOTO 575
945 PRINT
950 PRINT "IMPOSSIBLE THRUST-VALUE ";
955 IF F<0 THEN 985
960 IF F-.05<.05 THEN 975
965 PRINT "TOO LARGE"
970 GOTO 575
975 PRINT "TOO SMALL"
980 GOTO 575
985 PRINT "NEGATIVE"
990 GOTO 575
995 PRINT
1000 PRINT "TRANQUILITY BASE HERE -- THE EAGLE HAS LANDED"
1005 PRINT "CONGRATULATIONS - THERE WAS NO SPACECRAFT DAMAGE"
1010 PRINT "YOU MAY NOW PROCEED WITH SURFACE EXPLORATION."
1015 GOTO 1100
1020 PRINT
1025 PRINT "CRASH !!!!!!!!!!"
1030 PRINT "YOUR IMPACT CREATED A CRATER";ABS(H);M$;" DEEP"
1035 X1=SQR(D1*D1+H1*H1)*G3
1040 PRINT "AT CONTACT YOU WERE TRAVELLING";X1;N$;"/HR."
1045 GOTO 1100
1050 PRINT
1055 PRINT "YOU HAVE BEEN LOST IN SPACE WITH NO HOPE OF RECOVERY"
1060 GOTO 1100
1065 PRINT "YOU ARE DOWN SAFELY - "
1075 PRINT
1080 PRINT "BUT MISSED THE LANDING SITE BY";ABS(D/G5);N$
1085 GOTO 1100
1090 PRINT
1095 PRINT "MISSION ABORTED"
1100 PRINT
1105 PRINT "DO YOU WANT TO FLY IT AGAIN ? (YES OR NO)";
1110 INPUT Z$
1115 IF Z$="YES" THEN 20
1120 IF Z$="NO" THEN 1130
1125 GOTO 1105
1130 PRINT
1135 PRINT "TOO BAD, THE SPACE PROGRAM HATES TO LOSE EXPERIENCED";
1140 PRINT " ASTRONAUTS."
1145 STOP
1150 PRINT
1155 PRINT "OK, DO YOU WANT THE COMPLETE INSTRUCTIONS OR THE INPUT-"
1160 PRINT "OUTPUT STATEMENTS ?"
1165 PRINT "1=COMPLETE INSTRUCTIONS"
1170 PRINT "2=INPUT-OUTPUT STATEMENTS"
1175 PRINT "3=NEITHER"
1180 INPUT B1
1185 LET Q$="NO"
1190 IF B1=1 THEN 205
1195 LET Q$="YES"
1200 IF B1=2 THEN 190
1205 IF B1=3 THEN 190
1210 GOTO 1165
1215 END
```

ROCKT2 SAMPLE RUN

```
LUNAR LANDING SIMULATION

HAVE YOU FLOWN ON AN APOLLO/LEM MISSION BEFORE#(YES OR NO)? NO

WHICH SYSTEM OF MEASUREMENT DO YOU PREFER ?
 1=METRIC     0=ENGLISH
ENTER THE APPROPRIATE NUMBER? 0


 YOU ARE ON A LUNAR LANDING MISSION.  AS THE PILOT OF
THE LUNAR EXCURSION MODULE, YOU WILL BE EXPECTED TO
GIVE CERTAIN COMMANDS TO THE MODULE NAVIGATION SYSTEM.
 THE ON BOARD COMPUTER WILL GIVE A RUNNING ACCOUNT
OF INFORMATION NEEDED TO NAVIGATE THE SHIP.


THE ATTITUDE ANGLE CALLED FOR IS DESCRIBED AS FOLLOWS-
+ OR -180 DEGREES IS DIRECTLY AWAY FROM THE MOON
-90 DEGREES IS ON A TANGENT IN THE DIRECTION OF ORBIT
90 DEGREES IS ON A TANGENT FROM THE DIRECTION OF ORBIT
0 (ZERO) DEGREES IS DIRECTLY TOWARD THE MOON

                          -180,180
                             ↑
                    -90 < -+- > 90
                             !
                             0
                  << DIRECTION OF ORBIT <<

                      SURFACE OF MOON


ALL ANGLES BETWEEN -180 AND 180 DEGREES ARE ACCEPTED.

1 FUEL UNIT = 1 SEC. AT MAX. THRUST
ANY DISCREPANCIES ARE ACCOUNTED FOR IN THE USE OF FUEL
FOR AN ATTITUDE CHANGE.
AVAILABLE ENGINE POWER: 0 (ZERO) AND ANY VALUE BETWEEN
10 AND 100 PERCENT

NEGATIVE THRUST OR TIME IS PROHIBITED


INPUT: TIME INTERVAL IN SECONDS ------ (T)
       PERCENTAGE OF THRUST ---------- (P)
       ATTITUDE ANGLE IN DEGREES ----- (A)

FOR EXAMPLE:
T,P,A?10,65,-60
TO ABORT THE MISSION AT ANY TIME, ENTER 0,0,0

OUTPUT: TOTAL TIME ELAPSED IN SECONDS
        HEIGHT IN FEET
        DISTANCE FROM LANDING SITE IN FEET
        VERTICAL VELOCITY IN FEET/SECOND
        HORIZONTAL VELOCITY IN FEET/SECOND
        FUEL UNITS REMAINING

  0        364800       -1.928302E+7   0           5301.638   750
T,P,A? 20,20,-90
  20       364769.7     -1.918380E+7  -3.257229    5264.209   746.0001
 T,P,A? 200,10,-90
  220      358044.1     -1.821200E+7  -74.76607    5081.405   726.0002
 T,P,A? 500,10,-90
  720      224322.9     -1.589872E+7  -522.3451    4709.512   676.0002
 T,P,A? 500,0,0
  1040     -6773.7      -1.439553E+7  -918.728     4902.473   676.0002

CRASH !!!!!!!!!!
YOUR IMPACT CREATED A CRATER 6773.7 FEET DEEP
AT CONTACT YOU WERE TRAVELLING 2952.787  N.MIL/HR.

DO YOU WANT TO FLY IT AGAIN ? (YES OR NO)? YES

OK, DO YOU WANT THE COMPLETE INSTRUCTIONS OR THE INPUT-
OUTPUT STATEMENTS ?
1=COMPLETE INSTRUCTIONS
2=INPUT-OUTPUT STATEMENTS
3=NEITHER
? 3

ENTER MEASUREMENT OPTION NUMBER? 1

  0        111168       -5.876248E+6   0           1615.604   750
T,P,A? 500,0,0
  500      106291.7     -5.116247E+6  -19.20258    1619.915   750
T,P,A? 100,0,0
  600      104194.1     -4.963536E+6  -22.72435    1621.782   750
T,P,A? 50,90,-90
  650      102921.8     -4.892089E+6  -30.02382    1492.978   705.0001
T,P,A? 100,0,←→20,0
  750      101574.9     -4.749094E+6   3.226664    1494.091   685.0002
 T,P,A? 50,90,-90
  800      101326.7     -4.681730E+6  -14.90814    1359.475   640.0003
T,P,A? 100,40,-090
  900      97203.62     -4.558928E+6  -70.50644    1239.533   600.0003
T,P,A? 50,10,0
  950      93262.42     -4.530152E+6  -87.14224    1242.232   595.0004
 T,P,A? 50,100,0
  1000     92036.68     -4.441169E+6   38.9189     1243.085   545.0004
 T,P,A? 50,100,-90
  1050     93087.58     -4.386076E+6   1.363938    1079.071   495.0004
 T,P,A? 50,100,-90
  1100     92028.52     -4.338861E+6  -46.0884     910.9774   445.0004
T,P,A? 50,100,-90
  1150     88333.39     -4.299676E+6  -102.2833    738.2616   395.0005
T,P,A? 100,100,-90
  1250     71627.24     -4.246314E+6  -236.0606    375.7879   295.0005
T,P,A? 50,100,0
  1300     62820.57     -4.228179E+6  -115.1086    377.652    245.0005
 T,P,A? 50,100,0
  1350     60235.55     -4.209927E+6   12.98513    378.2053   195.0005
T,P,A? 100,50,0
  1450     64599.96     -4.173457E+6   75.8418     377.2816   145.0006
T,P,A? 100,40,-90
  1550     64756.08     -4.145499E+6  -73.66691    202.8112   105.0006
T,P,A? 50,50,-90
  1600     59156.86     -4.138397E+6  -150.5384    91.24105   80.0006
T,P,A? 10,0,0
  1610     57573.89     -4.137515E+6  -166.0619    91.32249   80.0006
T,P,A? 10,100,0
  1620     56062.7      -4.136631E+6  -136.1128    91.40043   70.00061
 T,P,A? 10,100,-90
  1630     54823.65     -4.135968E+6  -151.7089    45.55091   60.00062
T,P,A? 10,100,-90
  1640     53028.37     -4.135751E+6  -167.3554    -.746715   50.00063
 T,P,A? 10,0,0
  1650     51276.44     -4.135759E+6  -183.0355    -.7474552  50.00063
 T,P,A? 30,0,0
  1680     45077.51     -4.135780E+6  -230.2823    -.7500852  50.00063
T,P,A? 30,0,0
  1710     37455.8      -4.135802E+6  -277.9013    -.7533445  50.00063
 T,P,A? 50,0,0
  1760     21556.4      -4.135839E+6  -358.3198    -.7602341  50.00063
T,P,A? 10,100,0
  1770     18125.25     -4.135846E+6  -327.851     -.761739   40.00063
 T,P,A? 30,0,0
  1800     7553.268     -4.135869E+6  -377.0488    -.7664103  40.00063
 T,P,A? 10,50,0
  1810     3817.706     -4.135876E+6  -372.0578    -.7680755  35.00064
T,P,A? 5,100 ←,0
  1815     2005.734     -4.135880E+6  -354.7163    -.7688858  30.00064
 T,P,A? 5,100,0
  1820     270.6975     -4.135884E+6  -339.2829    -.7696634  25.00065
T,P,A? 1,100,0
  1820.8   .2621521     -4.135885E+6  -336.805     -.7697848  24.20065

CRASH !!!!!!!!!!
YOUR IMPACT CREATED A CRATER .2621521 METERS DEEP
AT CONTACT YOU WERE TRAVELLING 1212.501  KILOM/HR.

DO YOU WANT TO FLY IT AGAIN ? (YES OR NO)?  NO←←←NO

TOO BAD, THE SPACE PROGRAM HATES TO LOSE EXPERIENCED ASTRONAUTS.


READY
```