


COMPUTER GAMES

in BASIC



The four games in this booklet were written and tested on a PDP-10, using release 17 BASIC.

The games are:

Hide and Seek (Project Solo module #0201)	- page 1
NIM (Project Solo module #0202)	- page 6
MODULO (Project Solo module #0203)	- page 11
Space War (Project Solo module #0204)	- page 15

PROJECT SOLO / Computer Science Dept. / Univ. of Pittsburgh(15213)
Modules #0201-4 / Jeff Lederer (PS)

(Hide & Seek and MODULO were originated by students of mathematics teacher, Bud Valenti.)

HIDE AND SEEK

The following program is a game called "Hide and Seek". This game encourages students to become familiar with the Cartesian Co-ordinate system by asking them to find four players hidden at grid points on a ten by ten graph. Allowable grid points have the form (J,K) where J and K are integers between 0 and 9 inclusive. The student is given ten tries in which to find the player's position. After each try, he is told how close his guess was to each player. At the end of the game the student can ask to see where the players he didn't find were hidden. If the student wants to play another game the players move to new hidden locations.

After playing the game "in his head", the student should try to play the game with the aid of graph paper and a compass. A good student can find all the players in six or seven moves by "triangulation", using an approach that is similar to some radio navigational systems (e.g. LORAN).

A sample run of the program is given below. This is followed by a flow chart which explains the logic of the program. The numbers on the flow chart correspond to the statement numbers in the program listing. This listing is given on page 4. This program was written in BASIC for a PDP-10.

Sample RUN of /HIDE AND SEEK/

THIS IS THE GAME OF HIDE AND SEEK.

THE OBJECT OF THE GAME IS TO FIND THE FOUR PLAYERS
WHO ARE HIDDEN ON A 10 BY 10 GRID.

HOMEBASE WILL BE THE POSITION AT (0,0) AND ANY GUESS
YOU MAKE SHOULD CONTAIN TWO NUMBERS. THE FIRST GIVES
THE UNIT DISTANCE RIGHT OF THE HOMEBASE AND THE SECOND
IS THE UNIT DISTANCE ABOVE HOMEBASE.

YOU WILL HAVE TEN ATTEMPTS TO LOCATE THESE PLAYERS
AND WILL BE TOLD HOW CLOSE YOUR GUESS IS
TO EACH PLAYER.

IF AFTER 10 TRIES YOU ARE UNABLE TO CARRY OUT THIS TASK
YOU MAY CONTINUE TO BE 'IT', BUT THE PLAYERS WILL
BE PERMITTED TO MOVE TO NEW LOCATIONS.

ARE YOU READY TO BEGIN?
? YES

TURN NUMBER 1 , WHAT IS YOUR GUESS?
? 5,5

YOUR DISTANCE FROM PLAYER 1 IS 3.1 UNIT(S).
YOUR DISTANCE FROM PLAYER 2 IS 2.2 UNIT(S).
YOUR DISTANCE FROM PLAYER 3 IS 4.2 UNIT(S).
YOUR DISTANCE FROM PLAYER 4 IS 5.6 UNIT(S).

DO YOU WANT TO PLAY AGAIN?
? YES

2

TURN NUMBER 1 , WHAT IS YOUR GUESS?

? 5,5
YOUR DISTANCE FROM PLAYER 1 IS 5.6 UNIT(S).
YOUR DISTANCE FROM PLAYER 2 IS 5.8 UNIT(S).
YOUR DISTANCE FROM PLAYER 3 IS 4.4 UNIT(S).
YOUR DISTANCE FROM PLAYER 4 IS 3.1 UNIT(S).

TURN NUMBER 2 , WHAT IS YOUR GUESS?

? 8,0
YOUR DISTANCE FROM PLAYER 1 IS 11.4 UNIT(S).
YOUR DISTANCE FROM PLAYER 2 IS 11.3 UNIT(S).
YOUR DISTANCE FROM PLAYER 3 IS 9 UNIT(S).
YOUR DISTANCE FROM PLAYER 4 IS 8.2 UNIT(S).

TURN NUMBER 3 , WHAT IS YOUR GUESS?

? 8,9
YOUR DISTANCE FROM PLAYER 1 IS 7 UNIT(S).
YOUR DISTANCE FROM PLAYER 2 IS 8 UNIT(S).
YOUR DISTANCE FROM PLAYER 3 IS 1 UNIT(S).
YOUR DISTANCE FROM PLAYER 4 IS 2.2 UNIT(S).

TURN NUMBER 4 , WHAT IS YOUR GUESS?

? 7,9
YOUR DISTANCE FROM PLAYER 1 IS 6 UNIT(S).
YOUR DISTANCE FROM PLAYER 2 IS 7 UNIT(S).
YOU HAVE FOUND PLAYER 3
YOUR DISTANCE FROM PLAYER 4 IS 1.4 UNIT(S).

TURN NUMBER 5 , WHAT IS YOUR GUESS?

? 6,8
YOUR DISTANCE FROM PLAYER 1 IS 5 UNIT(S).
YOUR DISTANCE FROM PLAYER 2 IS 6 UNIT(S).
YOU HAVE FOUND PLAYER 4

TURN NUMBER 6 , WHAT IS YOUR GUESS?

? 0,8
YOUR DISTANCE FROM PLAYER 1 IS 1.4 UNIT(S).
YOU HAVE FOUND PLAYER 2

TURN NUMBER 7 , WHAT IS YOUR GUESS?

? 1,9
YOU HAVE FOUND PLAYER 1


YOU HAVE FOUND ALL THE PLAYERS IN 7 TURNS!

DO YOU WANT TO PLAY AGAIN?

? NO
THEN PLEASE LOGOUT.

Hide and Seek





3



```
5 RANDOMIZE
10 DIM P(4,2)
20 PRINT "THIS IS THE GAME OF HIDE AND SEEK."
30 PRINT
40 PRINT "THE OBJECT OF THE GAME IS TO FIND THE FOUR PLAYERS "
50 PRINT "WHO ARE HIDDEN ON A 10 BY 10 GRID."
60 PRINT
70 PRINT "HOMEBASE WILL BE THE POSITION AT (0,0) AND ANY GUESS "
80 PRINT "YOU MAKE SHOULD CONTAIN TWO NUMBERS. THE FIRST GIVES"
90 PRINT "THE UNIT DISTANCE RIGHT OF THE HOMEBASE AND THE SECOND"
100 PRINT "IS THE UNIT DISTANCE ABOVE HOMEBASE."
110 PRINT
120 PRINT "YOU WILL HAVE TEN ATTEMPTS TO LOCATE THESE PLAYERS"
130 PRINT "AND WILL BE TOLD HOW CLOSE YOUR GUESS IS"
140 PRINT "TO EACH PLAYER."
150 PRINT
160 PRINT "IF AFTER 10 TRIES YOU ARE UNABLE TO CARRY OUT THIS TASK"
170 PRINT "YOU MAY CONTINUE TO BE 'IT', BUT THE PLAYERS WILL"
180 PRINT "BE PERMITTED TO MOVE TO NEW LOCATIONS."
190 PRINT
200 PRINT
210 PRINT "ARE YOU READY TO BEGIN?"
220 INPUT AS
230 IF AS="NO" GOTO 620
240 GOSUB 1000
250 LET T=0
260 LET T=T+1
270 PRINT
280 PRINT
290 PRINT "TURN NUMBER",T," WHAT IS YOUR GUESS?"
300 INPUT M,N
310 FOR I=1 TO 4
320 IF P(I,1)=-1 GOTO 400
330 IF P(I,1)<>M GOTO 380
340 IF P(I,2)<>N GOTO 380
350 LET P(I,1)=-1
360 PRINT "YOU HAVE FOUND PLAYER",I
370 GOTO 400
380 LET D=SQR((P(I,1)-M)*2 + (P(I,2)-N)*2)
390 PRINT "YOUR DISTANCE FROM PLAYER",I,"IS",INT(D*10)/10,"UNIT(S)."
400 NEXT I
410 FOR J=1 TO 4
420 IF P(J,1)<>-1 GOTO 470
430 NEXT J
440 PRINT
450 PRINT "YOU HAVE FOUND ALL THE PLAYERS IN ",T," TURNS!"
460 GOTO 580
470 IF T<10 GO TO 260
480 PRINT
490 PRINT "YOU DIDN'T FIND ALL THE PLAYERS IN TEN TRIES."
500 PRINT "DO YOU WANT TO KNOW WHERE THE PLAYERS YOU DIDN'T "
510 PRINT "FIND WERE HIDDEN?"
520 INPUT BS
530 IF BS="NO" GOTO 580
```

```
540 FOR I=1 TO 4
550 IF P(I,1)=-1 GOTO 570
560 PRINT "PLAYER";I;" HID AT (";P(I,1);",";P(I,2);")."
570 NEXT I
580 PRINT
590 PRINT "DO YOU WANT TO PLAY AGAIN?"
600 INPUT C$
610 IF C$="YES" GOTO 240
620 PRINT "THEN PLEASE LOGOUT."
630 GOTO 9999
1000 FOR J=1 TO 2
1010 FOR I=1 TO 4
1020 P(I,J)=INT(RND*10)
1030 NEXT I
1040 NEXT J
1050 RETURN
9999 END
```


COMPUTER SIMULATIONS in BASIC



CONTENTS

NAME	MODULE #	AUTHOR(S)	PAGE
Lunar Landing	0221	Lawrence Siegel	1
Crazy Eights	0135	F. Wimberly & C. Len	5
Rectangular Billiards	9007	G. Lindstrom & C. Len	11
Elliptical Billiards	0113	G. Lindstrom & C. Len	25

PROJECT SOLO, Computer Science Department, University of Pittsburgh (15213)

LUNAR MODULE

3052 Warrington Road
Shaker Heights, Ohio 44120
May 30, 1972

Project Solo
University of Pittsburgh
Department of Computer Science
Pittsburgh, Pennsylvania 15213

Dear Sirs:

I thought you might be interested in a program I wrote recently. My program is called LMLAND. This means Lunar Module LANDer. It simulates the lunar module landing on the moon in only one direction (vertical). The user attempts to land the module with the following knowledge:

1. The LM starts at 70 miles from the surface.
2. The LM starts with 30,000 units of fuel.
3. Fuel is consumed at the rate of one unit for every 100 pounds of thrust per second.
4. The engine can be fired for 1,000 to 10,000 pounds of thrust for 10 second periods.
5. The engine does not have to be fired in a ten second period.
6. The thrust is entered in hundreds of pounds of thrust.

The program does not take into account the change in weight due to the change in height or the change in the amount of fuel (fuel weighs something!). The program tells the user what 10 second period he is working on, the rate downward in feet per second (fps), the number of miles from the surface, and the number of units of remaining fuel. After telling the user all of this information, the user must put in his thrust. If the number in the fps column becomes negative, the ship is traveling away from the surface of the moon. This simulator is fairly accurate in that the distance is realistic, the thrust available from the engine is realistic, and the calculations are accurate according to the mass of the lunar module, the gravitational pull, etc. The amount of fuel, and the fuel consumption are mere guesses that make the execution of the program interesting and a challenge. The ship must be landed at 30 fps or slower to be considered a landing. I landed the ship once. This is the only time it has ever been landed.

The program was written in BASIC for an HP 2116C.

Sincerely yours,

Lawrence Siegel

Lawrence Siegel
Ninth grade at Woodbury Junior High

At last--a module about modules. As can be seen from the above letter, this program was sent to us by a ninth grade student. We thought it was so good that we are mailing it to all of our readers. Our file number for this unit is #0221.

RUN
 YOU ARE IN THE LUNAR LANDER AND MUST LAND MANUALLY
 BECAUSE A SOLAR FLARE HAS ENDED COMMUNICATION WITH
 MISSION CONTROL. THE LM COMPUTER IS INOPERABLE.
 YOU HAVE 30,000 UNITS OF FUEL AND ARE 70 MILES FROM
 THE SURFACE. YOU CAN FIRE THE DESCENT ENGINE FOR 0 LBS.
 OF THRUST OR FOR 1000 TO 10000 LBS. OF THRUST FOR 10
 SECOND PERIODS. YOUR FUEL IS USED AT THE RATE OF
 1 UNIT FOR EVERY 100 POUNDS OF THRUST PER SECOND.
 THE THRUST IS ENTERED IN HUNDREDS OF POUNDS OF THRUST.
 GOOD LUCK

SEC	FPS	MILES UP	FUEL	THRUST(100#)
10	300	70	30000	?0
20	351.52	69.383	30000	?0
30	403.04	68.6685	30000	?-10
ILLEGAL ENTRY...TRY AGAIN				
30	403.04	68.6685	30000	?0
40	454.56	67.8564	30000	?10000
ILLEGAL ENTRY...TRY AGAIN				
40	454.56	67.8564	30000	?0
50	506.08	66.9467	30000	?0
60	557.6	65.9394	30000	?0
70	609.12	64.8345	30000	?0
80	660.64	63.6321	30000	?0
90	712.16	62.3321	30000	?0
100	763.68	60.9345	30000	?0
110	815.2	59.4394	30000	?0
120	866.72	57.8467	30000	?0
130	918.24	56.1564	30000	?0
140	969.76	54.3685	30000	?0
150	1021.28	52.483	30000	?0
160	1072.8	50.5	30000	?0
170	1124.32	48.4194	30000	?0
180	1175.84	46.2412	30000	?0
190	1227.36	43.9655	30000	?0
200	1278.88	41.5921	30000	?0
210	1330.4	39.1212	30000	?0
220	1381.92	36.5527	30000	?75
230	1357.97	33.9581	29250	?100
240	1308.87	31.4327	28250	?100
250	1259.76	29.0003	27250	?100
260	1210.66	26.6609	26250	?100
270	1161.55	24.4145	25250	?100
280	1112.45	22.2611	24250	?100
290	1063.34	20.2007	23250	?100
300	1014.24	18.2333	22250	?100
310	965.132	16.3589	21250	?100
320	916.027	14.5775	20250	?100
330	866.922	12.8891	19250	?100
340	817.817	11.2937	18250	?100
350	768.712	9.79127	17250	?100
360	719.607	8.38187	16250	?100
370	670.502	7.06548	15250	?100
380	621.397	5.84209	14250	?100
390	572.292	4.71171	13250	?100
400	523.187	3.67432	12250	?100
410	474.082	2.72994	11250	?100
420	424.977	1.87856	10250	?100
430	375.872	1.12018	9250	?100
440	326.767	.454799	8250	?100

RUN

YOU ARE IN THE LUNAR LANDER AND MUST LAND MANUALLY
BECAUSE A SOLAR FLARE HAS ENDED COMMUNICATION WITH
MISSION CONTROL. THE LM COMPUTER IS INOPERABLE.
YOU HAVE 30,000 UNITS OF FUEL AND ARE 70 MILES FROM
THE SURFACE. YOU CAN FIRE THE DESCENT ENGINE FOR 0 LBS.
OF THRUST OR FOR 1000 TO 10000 LBS. OF THRUST FOR 10
SECOND PERIODS. YOUR FUEL IS USED AT THE RATE OF
1 UNIT FOR EVERY 100 POUNDS OF THRUST PER SECOND.
THE THRUST IS ENTERED IN HUNDREDS OF POUNDS OF THRUST.
GOOD LUCK

SEC	FPS	MILES UP	FUEL	THRUST(100#)
10	300	70	30000	70
20	351.52	69.383	30000	70
30	403.04	68.6685	30000	70
40	454.56	67.8564	30000	70
50	506.08	66.9467	30000	70
. etc.				
200	1278.88	41.5921	30000	70
210	1330.4	39.1212	30000	775
220	1306.45	36.6242	29250	7100
230	1257.35	34.1963	28250	7100
240	1208.24	31.8615	27250	7100
250	1159.14	29.6197	26250	7100
260	1110.03	27.4708	25250	7100
270	1060.93	25.415	24250	7100
280	1011.82	23.4522	23250	7100
290	962.717	21.5824	22250	7100
300	913.612	19.8055	21250	7100
310	864.507	18.1217	20250	7100
320	815.402	16.5309	19250	7100
330	766.297	15.0331	18250	7100
340	717.192	13.6282	17250	7100
350	668.087	12.3164	16250	750
360	669.294	11.05	15750	7100
370	620.189	9.82886	14750	7100
380	571.084	8.70076	13750	7100
390	521.979	7.66566	12750	7100
400	472.875	6.72357	11750	7100
410	423.77	5.87447	10750	7100
420	374.665	5.11838	9750	7100
430	325.56	4.45529	8750	775
440	301.611	3.86138	8000	7100
450	252.506	3.33665	7000	750
460	253.713	2.85727	6500	775
470	229.765	2.39943	5750	7100
480	180.66	2.01077	4750	745
490	186.898	1.66271	4300	775
500	162.95	1.33141	3550	7100
510	113.845	1.0693	2550	7100
520	64.7395	.900183	1550	735
530	81.0408	.762134	1200	745
540	87.2795	.602739	750	745
550	93.5183	.431529	300	730

OUT OF FUEL AT .23421 MILES UP
IMPACT IN ABOUT 20 SECONDS
IMPACT 150.744 MPH YOU BLEW IT

```

20 PRINT "YOU ARE IN THE LUNAR LANDER AND MUST LAND MANUALLY"
30 PRINT "BECAUSE A SOLAR FLARE HAS ENDED COMMUNICATION WITH"
40 PRINT "MISSION CONTROL. THE LM COMPUTER IS INOPERABLE."
50 PRINT "YOU HAVE 30,000 UNITS OF FUEL AND ARE 70 MILES FROM"
60 PRINT "THE SURFACE. YOU CAN FIRE THE DESCENT ENGINE FOR 0 LBS."
70 PRINT "OF THRUST OR FOR 1000 TO 10000 LBS. OF THRUST FOR 10"
80 PRINT "SECOND PERIODS. YOUR FUEL IS USED AT THE RATE OF"
90 PRINT "1 UNIT FOR EVERY 100 POUNDS OF THRUST PER SECOND."
100 PRINT "THE THRUST IS ENTERED IN HUNDREDS OF POUNDS OF THRUST."
110 PRINT "GOOD LUCK"
180 PRINT
190 PRINT "SEC","FPS","MILES UP","FUEL","THRUST(100#)"
200 LET D=70
210 LET T=10
220 LET V=300
230 LET U=30000
240 PRINT T,V,D,U
250 INPUT F
260 IF F#INT(F) THEN 310
290 IF F>100 THEN 310
300 IF F*10>U THEN 310
305 IF F=0 THEN 330
308 IF F<10 THEN 310
309 GOTO 330
310 PRINT "ILLEGAL ENTRY...TRY AGAIN"
320 GOTO 240
330 LET U=U-F*10
340 LET V1=V
345 LET F=F*100
350 LET V2=(.16-F/32000)*322
360 LET D1=(V1+V2/2)/528
370 LET D=D-D1
380 LET V=V+V2
390 IF D <= 0 THEN 580
400 IF U <= 0 THEN 430
410 LET T=T+10
420 GOTO 240
430 PRINT "OUT OF FUEL AT"D;"MILES UP"
440 LET J=0
450 LET J=J+1
460 LET V1=V
470 LET V2=.16*332
480 LET D=D-(V1-V2/2)/528
485 LET V=V+V2
490 IF D>0 THEN 450
500 PRINT "IMPACT IN ABOUT"J*10;"SECONDS"
580 IF V <= 30 THEN 630
590 PRINT "IMPACT"V*15/22;"MPH YOU BLEW IT"
595 IF V<666 THEN 620
601 PRINT "THE SHIP WAS SABOTAGED"
620 STOP
630 PRINT "YOU MADE IT"
640 PRINT "TOUCH DOWN AT"V*15/22"MPH. YOU HAVE"U;"UNITS OF EXTRA"
650 PRINT "FUEL. BEFORE YOU REPORT TO NASA, PLEASE INFORM LARRY"
660 PRINT "SIEGEL, 3052 WARRINGTON RD., SHAKER HTS., OHIO 44120,"
665 PRINT "AS TO HOW YOU SUCCEEDED. CONGRATULATIONS"
670 END

```