

Psychoanalysis (?) by Computer...

ELIZA

Steve North

Language: BASIC (MITS 8K)

Author: Originally programmed in LISP by Joseph Weizenbaum. This version is based on one written by Jeff Shrager.

Description: ELIZA is a program that accepts natural English as input and carries on a reasonably coherent conversation based on the psychoanalytic techniques of Carl Rogers. You will have to forgive ELIZA for being a poor English student. You'll find that it is best not to use punctuation in your input, and you'll have to carry the conversation. But it does work!

Remarks: In order to do what it does, ELIZA must: (1) get a string from the user, and prepare it for further processing; (2) find the keywords in the input string; (3) if a keyword is found; take the part of the string following the keyword and "translate" all the personal pronouns and verbs ("I" becomes "YOU", "ARE" becomes "AM", etc.); (4) finally, look up an appropriate reply based on the keyword which was found, print it and, if necessary, the "translated" string. ELIZA uses four types of program data to accomplish this:

(1) 36 keyword, such as "I AM", "WHY DONT YOU", and "COMPUTER". The keywords must be in order of priority, so ELIZA will key on "YOU ARE" before "YOU".


(2) 12 strings used for the translation or conjugation process. These are in pairs

Sample Run

```

RUN
HI! I'M ELIZA. WHAT'S YOUR PROBLEM?
? I AM AN IBM 370
DID YOU COME TO ME BECAUSE YOU ARE AN IBM 370
? YES I WANT A NEW LINEPRINTER
WHAT WOULD IT MEAN TO YOU IF YOU GOT A NEW LINEPRINTER
? ABOUT 1200 LPM
SAY, DO YOU HAVE ANY PSYCHOLOGICAL PROBLEMS?
? WHAT WOULD YOU KNOW I'M A 370 AND YOU'RE AN IMSAI
WHAT MAKES YOU THINK I AM AN IMSAI
? IT SAYS THAT ON YOUR FRONT PANEL
WHY ARE YOU CONCERNED ABOUT MY FRONT PANEL
? LOOK YOU BROUGHT UP THE SUBJECT
WE WERE DISCUSSING YOU-- NOT ME.
? I'M FIXING A HOLE...
DID YOU COME TO ME BECAUSE YOU ARE FIXING A HOLE...
? SORRY THAT WAS A TEN YEAR OLD SOFTWARE BUG
PLEASE DON'T APOLOGIZE!
? I AM ALWAYS APOLOGIZING
HOW LONG HAVE YOU BEEN ALWAYS APOLOGIZING
? SINCE MY FIRST COLD START
WHAT DOES THAT SUGGEST TO YOU?
? YOU'RE THE SHRINK
DOES IT PLEASE YOU TO BELIEVE I AM THE SHRINK
? DONT YOU KNOW?
OH, I KNOW?
? I AM BEGINNING TO WONDER ABOUT YOU
DO YOU BELIEVE IT IS NORMAL TO BE BEGINNING TO WONDER ABOUT I
? HA HA HA YOU CAN'T EVEN TALK PROPERLY
YOU'RE NOT REALLY TALKING ABOUT ME, ARE YOU?
? IS THERE ANYONE ELSE IN THE ROOM?
I SEE.
? COMPUTERS DONT HAVE EYES
ARE YOU SURE?
? ABOUT COMPUTERS?
DO COMPUTERS WORRY YOU?
? SHOULD THEY AFTER ALL I AM A 370
DO YOU ENJOY BEING A 370
?

```



```

10 REM
20 REM      ELIZA/DOCTOR
30 REM      CREATED BY JOSEPH WEIZENBAUM
40 REM      THIS VERSION BY JEFF SHRAGER
50 REM      EDITED AND MODIFIED FOR MITS 8K BASIC 4.0 BY STEVE NORTH
60 REM      CREATIVE COMPUTING PO BOX 789-M MORRISTOWN NJ 07960
70 REM
80 REM      -----INITIALIZATION-----
90 DIM C$(72),I$(72),K$(72),F$(72),S$(72),R$(72),P$(72),Z$(72)
100 DIM S(36),R(36),N(36)
110 N1=36:N2=12:N3=112
120 FOR X=1 TO N1+N2+N3:READ Z$:NEXT X:REM SAME AS RESTORE
130 FORX=1 TO N1
140 READ S(X),L:R(X)=S(X):N(X)=S(X)+L-1
150 NEXT X
160 PRINT "HI! I'M ELIZA. WHAT'S YOUR PROBLEM?"
170 REM
180 REM      -----USER INPUT SECTION-----
190 REM
200 INPUT I$
201 I$=" "+I$+" "
210 REM GET RID OF APOSTROPHES
220 FOR L=1 TO LEN(I$)
230 IF MID$(I$,L,1)="'" THEN I$=LEFT$(I$,L-1)+RIGHT$(I$,LEN(I$)-L):GOTO 230
240 IFL+4<=LEN(I$) THEN IF MID$(I$,L,4)="SHUT" THEN PRINT "SHUT UP...":END

```

Program Listing

such that if one member of the pair is found, the other is substituted for it. Examples: "Y", "YOU", "AM", "ARE", etc.

(3) 112 reply strings. The strings are arranged in groups corresponding to the keywords. There is no fixed number of different replies for each keyword. Replies ending in a "*" are to be followed by the translated string, while the strings ending in normal punctuation are to be printed alone.

(4) Numerical data to determine which replies to print for each keyword. For each keyword there is a pair of numbers signifying (start of reply strings, number of reply strings). Thus the fifth pair of number, (10,4), means that the replies for the fifth keyword ("I DONT") start with the tenth reply string, and that there are four replies.

Detailed Explanation:

Lines 10-160: Initialization. Arrays and strings are dimensioned. N1, N2, and N3, which represent the number of keywords, number of translation strings, and number of replies respectively, are defined. Then the arrays are filled. S(keyword number) is the ordinal number of the start of the reply strings for a given keyword, R(keyword number) is the actual reply to be used next, and N(keyword number) is the last reply for that keyword. Finally an introduction is printed.

Lines 170-255: User input section. This part of the program gets a string from the user, places a space at the start of the string and two at the end (to make it easier to correctly locate keywords and to prevent subscripting out of bounds), throws out all the apostrophes (so DONT and DON'T are equivalent), and stops if the word SHUT is found in the input string (which it takes to mean SHUT UP). ELIZA also checks for repetitive input by the user.

Lines 260-370: Keyword-finding section. ELIZA scans the input string for keywords and saves the keyword of highest priority temporarily in S, T, and F\$. If no keyword is found, the keyword defaults to number 36, NOKEYFOUND (which causes ELIZA to say something noncommittal) and it skips the next section.

Lines 380-555: Translation or Conjugation section. The part of the input string following the keyword is saved. Then pairs of translation strings, as described above, are read and upon the occurrence of one of these strings, the other is substituted for it. When this is done ELIZA makes sure there is only one leading space in the translated string.

Lines 560-640: Reply printing section. Using R(keyword number), S(keyword number), and N(keyword number), the correct reply is located. The pointer for the next reply is bumped and reset if it

```

250 NEXT L
255 IF I$=P$ THEN PRINT "PLEASE DON'T REPEAT YOURSELF!":GOTO 170
260 REM
270 REM -----FIND KEYWORD IN I$-----
280 REM
290 RESTORE
295 S=0
300 FOR K=1 TO N1
310 READ K$
315 IF S>0 THEN 360
320 FOR L=1 TO LEN(I$)-LEN(K$)+1
340 IF MID$(I$,L,LEN(K$))=K$ THEN S=K:T=L:F$=K$
350 NEXT L
360 NEXT K
365 IF S>0 THEN K=S:L=T:GOTO 390
370 K=36:GOTO 570:REM WE DIDN'T FIND ANY KEYWORDS
380 REM
390 REM TAKE RIGHT PART OF STRING AND CONJUGATE IT
400 REM USING THE LIST OF STRINGS TO BE SWAPPED
410 REM
420 RESTORE:FOR X=1 TO N1:READ Z$:NEXT X:REM SKIP OVER KEYWORDS
430 C$=" "+RIGHT$(I$,LEN(I$)-LEN(F$)-L+1)
440 FOR X=1 TO N2/2
450 READ S$,R$
460 FOR L=1 TO LEN(C$)
470 IF L+LEN(S$)>LEN(C$) THEN 510
480 IF MID$(C$,L,LEN(S$))<>S$ THEN 510
490 C$=LEFT$(C$,L-1)+R$+RIGHT$(C$,LEN(C$)-L-LEN(S$)+1)
495 L=L+LEN(R$)
500 GOTO 540
510 IF L+LEN(R$)>LEN(C$) THEN 540
520 IF MID$(C$,L,LEN(R$))<>R$ THEN 540
530 C$=LEFT$(C$,L-1)+S$+RIGHT$(C$,LEN(C$)-L-LEN(R$)+1)
535 L=L+LEN(S$)
540 NEXT L
550 NEXT X
555 IF MID$(C$,2,1)=" " THEN C$=RIGHT$(C$,LEN(C$)-1):REM ONLY 1 SPACE
560 REM
570 REM NOW USING THE KEYWORD NUMBER (K) GET REPLY
580 REM
590 RESTORE:FOR X=1 TO N1+N2:READ Z$:NEXT X
600 FOR X=1 TO R(K):READ F$:NEXT X:REM READ RIGHT REPLY
610 R(K)=R(K)+1:IF R(K)>N(K) THEN R(K)=S(K)
620 IF RIGHT$(F$,1)<>"*" THEN PRINT F$:P$=I$:GOTO 170
630 PRINT LEFT$(F$,LEN(F$)-1);C$
640 P$=I$:GOTO 170
1000 REM
1010 REM -----PROGRAM DATA FOLLOWS-----
1020 REM
1030 REM KEYWORDS
1040 REM
1050 DATA "CAN YOU","CAN I","YOU ARE","YOU'RE","I DONT","I FEEL"
1060 DATA "WHY DONT YOU","WHY CANT I","ARE YOU","I CANT","I AM","IM "
1070 DATA "YOU ","I WANT","WHAT","HOW","WHO","WHERE","WHEN","WHY"
1080 DATA "NAME","CAUSE","SORRY","DREAM","HELLO","HI ","MAYBE"
1090 DATA " NO","YOUR","ALWAYS","THINK","ALIKE","YES","FRIEND"
1100 DATA "COMPUTER","NOKEYFOUND"
1200 REM
1210 REM STRING DATA FOR CONJUGATIONS
1220 REM
1230 DATA " ARE "," AM "," WERE "," WAS "," YOU "," I "," YOUR "," MY "
1235 DATA " IVE "," YOUVE "," IM "," YOURE "
1300 REM
1310 REM REPLIES
1320 REM
1330 DATA "DON'T YOU BELIEVE THAT I CAN*"
1340 DATA "PERHAPS YOU WOULD LIKE TO BE ABLE TO*"
1350 DATA "YOU WANT ME TO BE ABLE TO*"
1360 DATA "PERHAPS YOU DON'T WANT TO*"
1365 DATA "DO YOU WANT TO BE ABLE TO*"
1370 DATA "WHAT MAKES YOU THINK I AM*"
1380 DATA "DOES IT PLEASE YOU TO BELIEVE I AM*"
1390 DATA "PERHAPS YOU WOULD LIKE TO BE*"
1400 DATA "DO YOU SOMETIMES WISH YOU WERE*"
1410 DATA "DON'T YOU REALLY*"
1420 DATA "WHY DON'T YOU*"
1430 DATA "DO YOU WISH TO BE ABLE TO*"
1440 DATA "DOES THAT TROUBLE YOU?"
1450 DATA "TELL ME MORE ABOUT SUCH FEELINGS."
1460 DATA "DO YOU OFTEN FEEL*"
1470 DATA "DO YOU ENJOY FEELING*"
1480 DATA "DO YOU REALLY BELIEVE I DON'T*"
1490 DATA "PERHAPS IN GOOD TIME I WILL*"
1500 DATA "DO YOU WANT ME TO*"
1510 DATA "DO YOU THINK YOU SHOULD BE ABLE TO*"
1520 DATA "WHY CAN'T YOU*"
1530 DATA "WHY ARE YOU INTERESTED IN WHETHER OR NOT I AM*"
1540 DATA "WOULD YOU PREFER IF I WERE NOT*"
1550 DATA "PERHAPS IN YOUR FANTASIES I AM*"

```

too large. If the reply string ends in a "*" it is printed with the translated string, otherwise it is printed alone. The previously entered input string is saved to permit checking for repetitive input, and then ELIZA goes back for more input.

Limitations: Runs in 16K of memory.

Modifications: You can easily add, change, or delete any of the keywords, translation words, or replies. Remember, you will also have to change N1, N2, N3, and/or the numerical data. Just as a suggestion, if you decide to insert "ME" and "YOU" in the translation string list, put a nonprinting (control) character in YOU to prevent ELIZA from substituting I-YOU-ME. This means that YOU will always be assumed to be the subject of a verb, never the object, but resolving that difficulty is a whole different problem.

A Few Comments: The structures found in lines 120, 420, and 590 could be replaced by RESTORE NNNN statements if your BASIC has them. The use of an INSTR, SEARCH, or POS function to determine if one string is a substring of another would probably speed things up considerably (it takes ELIZA around 10 seconds to think of a reply).

What it all means: we'll leave to you. Although this program is an inferior imitation of the original, it does work. It is pretty farfetched to believe that a psychanalyst is nothing but a sentence-input-keyword-finder-conjugator-reply finder, but if you really think so, you can buy your computer a speech-recognition unit, a Computalker and a green couch, and charge \$75/hr! My computer, the doctor!

MITS STRING BASIC FUNCTIONS

LEFT\$(X\$,X) takes the X leftmost characters of X\$.

RIGHT\$(X\$,X) takes the X rightmost characters of X\$.

MID\$(X\$,X,Y) takes Y characters from X\$ starting with the Xth character.

DIM X\$(72) dimensions X\$ to hold 72 characters. That may seem obvious, but there's a story that goes along with this. The listing of ELIZA was made on a friend's SWTPC 6800 running 8K BASIC. After making a small patch in BASIC, loading a paper tape, and changing all the multiple statements to separate lines, we were all ready to try out ELIZA when Bob (whose computer it was asked about the enormous strings ELIZA used. It turns out that in SWTPC BASIC, DIM X\$(72) means dimension string vector X\$ to have 72 little strings(X\$(K)) of 18 character

```

1560 DATA "HOW DO YOU KNOW YOU CAN'T*"
1570 DATA "HAVE YOU TRIED?"
1580 DATA "PERHAPS YOU CAN NOW*"
1590 DATA "DID YOU COME TO ME BECAUSE YOU ARE*"
1600 DATA "HOW LONG HAVE YOU BEEN*"
1610 DATA "DO YOU BELIEVE IT IS NORMAL TO BE*"
1620 DATA "DO YOU ENJOY BEING*"
1630 DATA "WE WERE DISCUSSING YOU-- NOT ME."
1640 DATA "OH, I*"
1650 DATA "YOU'RE NOT REALLY TALKING ABOUT ME, ARE YOU?"
1660 DATA "WHAT WOULD IT MEAN TO YOU IF YOU GOT*"
1670 DATA "WHY DO YOU WANT*"
1680 DATA "SUPPOSE YOU SOON GOT*"
1690 DATA "WHAT IF YOU NEVER GOT*"
1700 DATA "I SOMETIMES ALSO WANT*"
1710 DATA "WHY DO YOU ASK?"
1720 DATA "DOES THAT QUESTION INTEREST YOU?"
1730 DATA "WHAT ANSWER WOULD PLEASE YOU THE MOST?"
1740 DATA "WHAT DO YOU THINK?"
1750 DATA "ARE SUCH QUESTIONS ON YOUR MIND OFTEN?"
1760 DATA "WHAT IS IT THAT YOU REALLY WANT TO KNOW?"
1770 DATA "HAVE YOU ASKED ANYONE ELSE?"
1780 DATA "HAVE YOU ASKED SUCH QUESTIONS BEFORE?"
1790 DATA "WHAT ELSE COMES TO MIND WHEN YOU ASK THAT?"
1800 DATA "NAMES DON'T INTEREST ME."
1810 DATA "I DON'T CARE ABOUT NAMES-- PLEASE GO ON."
1820 DATA "IS THAT THE REAL REASON?"
1830 DATA "DON'T ANY OTHER REASONS COME TO MIND?"
1840 DATA "DOES THAT REASON EXPLAIN ANYTHING ELSE?"
1850 DATA "WHAT OTHER REASONS MIGHT THERE BE?"
1860 DATA "PLEASE DON'T APOLOGIZE!"
1870 DATA "APOLOGIES ARE NOT NECESSARY."
1880 DATA "WHAT FEELINGS DO YOU HAVE WHEN YOU APOLOGIZE."
1890 DATA "DON'T BE SO DEFENSIVE!"
1900 DATA "WHAT DOES THAT DREAM SUGGEST TO YOU?"
1910 DATA "DO YOU DREAM OFTEN?"
1920 DATA "WHAT PERSONS APPEAR IN YOUR DREAMS?"
1930 DATA "ARE YOU DISTURBED BY YOUR DREAMS?"
1940 DATA "HOW DO YOU DO ... PLEASE STATE YOUR PROBLEM."
1950 DATA "YOU DON'T SEEM QUITE CERTAIN."
1960 DATA "WHY THE UNCERTAIN TONE?"
1970 DATA "CAN'T YOU BE MORE POSITIVE?"
1980 DATA "YOU AREN'T SURE?"
1990 DATA "DON'T YOU KNOW?"
2000 DATA "ARE YOU SAYING NO JUST TO BE NEGATIVE?"
2010 DATA "YOU ARE BEING A BIT NEGATIVE."
2020 DATA "WHY NOT?"
2030 DATA "ARE YOU SURE?"
2040 DATA "WHY NOT?"
2050 DATA "WHY ARE YOU CONCERNED ABOUT MY*"
2060 DATA "WHAT ABOUT YOUR OWN*"
2070 DATA "CAN YOU THINK OF A SPECIFIC EXAMPLE?"
2080 DATA "WHEN?"
2090 DATA "WHAT ARE YOU THINKING OF?"
2100 DATA "REALLY, ALWAYS?"
2110 DATA "DO YOU REALLY THINK SO?"
2120 DATA "BUT YOU ARE NOT SURE YOU*"
2130 DATA "DO YOU DOUBT YOU*"
2140 DATA "IN WHAT WAY?"
2150 DATA "WHAT RESEMBLANCE DO YOU SEE?"
2160 DATA "WHAT DOES THE SIMILARITY SUGGEST TO YOU?"
2170 DATA "WHAT OTHER CONNECTIONS DO YOU SEE?"
2180 DATA "COULD THERE REALLY BE SOME CONNECTION?"
2190 DATA "HOW?"
2200 DATA "YOU SEEM QUITE POSITIVE."
2210 DATA "ARE YOU SURE?"
2220 DATA "I SEE."
2230 DATA "I UNDERSTAND."
2240 DATA "WHY DO YOU BRING UP THE TOPIC OF FRIENDS?"
2250 DATA "DO YOUR FRIENDS WORRY YOU?"
2260 DATA "DO YOUR FRIENDS PICK ON YOU?"
2270 DATA "ARE YOU SURE YOU HAVE ANY FRIENDS?"
2280 DATA "DO YOU IMPOSE ON YOUR FRIENDS?"
2290 DATA "PERHAPS YOUR LOVE FOR FRIENDS WORRIES YOU."
2300 DATA "DO COMPUTERS WORRY YOU?"
2310 DATA "ARE YOU TALKING ABOUT ME IN PARTICULAR?"
2320 DATA "ARE YOU FRIGHTENED BY MACHINES?"
2330 DATA "WHY DO YOU MENTION COMPUTERS?"
2340 DATA "WHAT DO YOU THINK MACHINES HAVE TO DO WITH YOUR PROBLEM?"
2350 DATA "DON'T YOU THINK COMPUTERS CAN HELP PEOPLE?"
2360 DATA "WHAT IS IT ABOUT MACHINES THAT WORRIES YOU?"
2370 DATA "SAY, DO YOU HAVE ANY PSYCHOLOGICAL PROBLEMS?"
2380 DATA "WHAT DOES THAT SUGGEST TO YOU?"
2390 DATA "I SEE."
2400 DATA "I'M NOT SURE I UNDERSTAND YOU FULLY."
2410 DATA "COME COME ELUCIDATE YOUR THOUGHTS."
2420 DATA "CAN YOU ELABORATE ON THAT?"
2430 DATA "THAT IS QUITE INTERESTING."
2500 REM

```


each. There's no easy way to handle more than 18 characters at a time! Those of you using SWTPC 6800 BASIC should have a lot of fun with this.

```

2510 REM      DATA FOR FINDING RIGHT REPLIES
2520 REM
2530 DATA 1,3,4,2,6,4,6,4,10,4,14,3,17,3,20,2,22,3,25,3
2540 DATA 28,4,28,4,32,3,35,5,40,9,40,9,40,9,40,9,40,9
2550 DATA 49,2,51,4,55,4,59,4,63,1,63,1,64,5,69,5,74,2,76,4
2560 DATA 80,3,83,7,90,3,93,6,99,7,106,6

```

Name	Usage
R(X),S(X),N(X)	See Text
IS	Input string
K\$	Keyword string
C\$	Translated or conjugated string
F\$	Reply string, also used to save K\$ in scanning for keyword
R\$,S\$	Strings used in conjugation process
P\$	Previous input string
Z\$	Scratch (used for simulating RESTORE NNNN statement).
N1	Number of keywords
N2	Number of conjugation strings
N3	Number of replies
K	Keyword number
S,T	Used to save K and L when scanning for keyword
X,L	X,L Scratch. X is generally used for looping while L is used for scanning through strings
V	V Used for scanning for keyword string



DATA PROCESSING DEFINITIONS

ASSUMED DECIMAL POINT. Located two positions to the right of a programmer's current salary in estimating his own worth.

BIT. The increment by which programmers slowly go mad.

CHAINING. A method of attaching programmers to desks to speed up output.

CHECKPOINT. The location from which a programmer draws his salary.

COMMON LANGUAGE. The first thing a programmer must forget in order to be successful.

CORE STORAGE. A receptacle for the center section of apples.

COUNTER. A device over which martinis are served.

ERROR. What someone else has made when he disagrees with your computer output.

EXTERNAL STORAGE. Wastebasket.

FIXED WORD LENGTH. Four-letter words used by programmers in a state of confusion.

FLOATING CONTROL. A characteristic exhibited when you have to go to the restroom but can't leave the computer.

FLOATING POINT. The absolute limit before floating control is lost.

FLOW CHART. A graphic representation of the fastest route to the restroom.

INPUT. Food, whiskey, beer, aspirin, etc.

MACRO. The last half of an expression of surprise: "Holy Macro."

MEMORY DUMP. Amnesia.

PROGRAMMER. Red-eyed, mumbling mammal capable of conversing with inanimate objects.


—Modern Data

The COMPUTALKER Model CT-1 optimizes the trade-off between low data rate speech and directly digitized speech. Low data rate speech relies on canned definitions for the sound of each phoneme, which produces mechanical sounding speech. Digitized speech, while remaining faithful to the original sound, requires 10K to 20K bytes per second of storage and is inflexible to phonetic manipulation.

With the Model CT-1, the sounds are defined in real time under software control. Parameters which represent the phonetic structure of human speech are transmitted to the CT-1 at a rate of 500 to 900 bytes per second. This allows the production of highly intelligible and quite natural sounding speech output. Speaker characteristics and language or dialect variations are retained in the output.

The CT-1 can also be operated in a low data rate mode using phoneme definitions contained in the CSR1 Synthesis-by-Rule software package. The COMPUTALKER speech synthesis system, used in this way, has the advantage that the software driver can easily be modified to keep the naturalness and intelligibility of the speech output up to date with the constantly evolving state of the art of rule governed speech.

COMPUTALKER



SPEAK "KAAMPYUWTAOLKER"

MODEL CT-1 SYNTHESIZER	395.00
CSR1 SOFTWARE SYSTEM	35.00
DEMONSTRATION CASSETTE	2.95

CALIF. RESIDENTS ADD 6% SALES TAX

COMPUTALKER CONSULTANTS
Box 1951-E, Santa Monica, CA 90406

CALIFORNIA:

Long Beach, CA 90815
A-VIDD Electronics
[213] 598-0444
Lawndale, CA 90260
Byte Shop of Lawndale
[213] 371-2421
Orange, CA 92667
The Computer Mart
[714] 633-1222
Santa Monica, CA 90401
The Computer Store
[213] 451-0713

GEORGIA:

Atlanta, GA 30305
Computer Systems Center
[404] 231-1691

INDIANA:

Bloomington, IN 47401
The Data Domain
[812] 334-3607

KANSAS:

Kansas City, MO 64113
Micro-Com, Inc.
[816] 333-8383

MICHIGAN:

Royal Oak, MI 48073
Computer Mart of Royal Oak
[313] 576-0900

NEW JERSEY:

Hoboken, NJ 07030
Hoboken Computer Works
[201] 420-1644

OREGON:

Portland, OR 97201
Byte Shop of Portland
[503] 223-3496

VIRGINIA, WASH. D.C.:

Fairfax, VA 22030
Roberts Information Services, Inc.
[703] 560-6119

CANADA:

Toronto, Ontario, Canada
M5V 1Z1
The Computer Place
[416] 598-0262

ENGLAND:

London NW6, England
Compelec Electronics

GERMANY, AUSTRIA:

8011 Munich-Putzbrunn
Datameg Div. of Megatron KG
[089] 46 50 66

JAPAN:

Shibuya-ku, Tokyo, Japan
Pax Electronics
[03] 370-2751