4K RAM

AREA: Games/Physics NUMBER: NAME: LUNAR
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: The program represents an exact simulation of an Apollo lunar landing module during the final descent. This portion of the descent would normally be controlled by the on-board computer backed up by a computer located on Earth. However, to exercise your knowledge of physics (and make an interesting game), both computers simultaneously have had a malfunction; therefore, you are on your own to safely land the spacecraft.

To make a soft landing, you may reset the burn rate of the retro rockets every ten seconds. You have a choice of not firing at all (burn rate=0) or firing at a rate between 8 and 200 lbs. per second. You have 16500 lbs. of fuel. If the rockets were not fired, your estimated free fall impact time is 120 seconds. The capsule weight is 33,000 pounds

USAGE: This program runs on EduSystem 20 and larger systems. Type SCR to get rid of any existing programs and load LUNAR. Type RUN to commence landing procedure. When a "?" is typed under burn rate, enter your burn rate and hit the return key. Remember, the only acceptable burn rates are 0 or any number between 8 and 200.

After you're down and the computer has typed "READY", if you wish to try it again, simply type RUN. Good luck!

```
RUN
GROUND CONTROL CALLING LUNAR MODULE.....
ON-BOARD AND GROUND COMPUTERS HAVE SIMULTANEOUSLY
MALFUNCTIONED  (THEY WEREN'T DEC MACHINES)
MANUAL CONTROL IS NECESSARY.

CAPSULE WEIGHT 32,500 LBS    AVAILABLE FUEL 16,500 LBS
ESTIMATED FREE FALL IMPACT TIME 120 SECONDS

SET RETRO ROCKET BURN RATE EVERY 10 SECONDS TO ANY VALUE
BETWEEN 0 LBS/SEC (FREE FALL) AND
200 LBS/SEC (STRONG BRAKING)

GOOD LUCK   !!!

SEC             MI + FT        MPH             LB FUEL        BURN RATE

 0               120  0         3600            16500          ? 0
 10              109  5016      3636            16500          ? 0
 20              99   4224      3672            16500          ? 0
 30              89   2904      3708            16500          ? 0
 40              79   1056      3744            16500          ? 0
 50              68   3960      3780            16500          ? 0
 60              58   1056      3816            16500          ? 0
 70              47   2904      3852            16500          ? 200
 80              37   1884      3482.868        14500          ? 200
 90              28   1191      3086.708        12500          ? 200
 100             20   1251      2659.654        10500          ? 200
 110             13   2549      2196.947        8500           ? 200
 120             8    370       1692.634        6500           ? 200
 130             4    658       1139.138        4500           ? 200
 140             1    4204      526.5976        2500           ? 100
 150             0    4042      212.242         1500           ? 45
 160             0    1864      84.18261        1050           ? 20
 170             0    909       45.91246        850            ? 17
 180             0    438       18.10655        680            ? 13
 190             0    269       4.886334        550            ? 10
 200             0    213       2.768577        450            ? 10
 210             0    189       .4252701        350            ? 9.2
 220             0    179       .9479145        258            ? 9.3
 230             0    165       .8864265        165            ? 9.4
 240             0    157       .2320987        71             ? 7.1
RAN OUT OF FUEL AT 250 SEC
ON THE MOON AT 254.0616 SEC - IMPACT VELOCITY 23.03008 MPH
CRAFT DAMAGE.  YOU'RE STRANDED HERE UNTIL A RESCUE MISSION
ARRIVES.  HOPE YOUR OXYGEN HOLDS OUT!

READY
```

```
READY

TAPE

    1 PRI "GROUND CONTROL CALLING LUNAR MODULE....."
    2 PRI "ON-BOARD AND GROUND COMPUTERS HAVE SIMULTANEOUSLY"
    3 PRI "MALFUNCTIONED  (THEY WEREN'T DEC MACHINES)"
    4 PRI "MANUAL CONTROL IS NECESSARY."
    5 PRI
    6 PRI "CAPSULE WEIGHT 32,500 LBS    AVAILABLE FUEL 16,500 LBS"
    7 PRI "ESTIMATED FREE FALL IMPACT TIME 120 SECONDS"
    8 PRI
    9 PRI "SET RETRO ROCKET BURN RATE EVERY 10 SECONDS TO ANY VALUE"
   10 PRI "BETWEEN 0 LBS/SEC (FREE FALL) AND"
   11 PRI "200 LBS/SEC (STRONG BRAKING)"
   12 PRI
   13 PRI "GOOD LUCK   !!!"
   14 PRI
   15 PRI "SEC","MI + FT","MPH","LB FUEL","BURN RATE"
   16 PRI
   20 A=120
   21 V=1
   22 M=33000
   23 N=16500
   24 G=.001
   25 Z=1.8
   30 PRI L,INT(A);INT(5280*(A-INT(A))),3600*V,M-N,
   31 INPUT K
   32 T=10
   40 IF M-N<.001 THEN 70
   41 IF T<.001 THEN 30
   42 S=T
   43 IF M>=N+S*K THEN 50
   44 S=(M-N)/K
   50 GOSUB 170
   51 IF I<=0 THEN 130
   52 IF V<=0 THEN 60
   53 IF J<0 THEN 150
   60 GOSUB 120
   61 GOTO 40
   70 PRINT "RAN OUT OF FUEL AT"L"SEC"
   71 S=(-V+SQR(V*V+2*A*G))/G
   72 V=V+G*S
   73 L=L+S
   80 W=3600*V
   81 PRINT "ON THE MOON AT"L"SEC - IMPACT VELOCITY"W"MPH"
   82 IF W>1.2 THEN 90
   83 PRI "PERFECT LANDING!  (LUCKY)"
   84 STOP
   90 IF W>10 THEN 100
   91 PRI "GOOD LANDING  (COULD BE BETTER)"
   92 STOP
  100 IF W>60 THEN 110
  101 PRI "CRAFT DAMAGE.  YOU'RE STRANDED HERE UNTIL A RESCUE MISSION"
  102 PRI "ARRIVES.  HOPE YOUR OXYGEN HOLDS OUT!"
  103 STOP
  110 PRI "SORRY, BUT THERE WERE NO SURVIVORS -- YOU BLEW IT."
  111 PRI "IN FACT, YOU BLASTED A NEW LUNAR CRATER"W*.2777"FT DEEP"
  112 STOP
```

```
215  120 L=L+S
220  121 T=T-S
225  122 M=M-S*K
230  123 A=I
235  124 V=J
240  125 RET
245  130 IF S<.005 THEN 80
250  131 D=V+SQR(V*V+2*A*(G-Z*K/M))
255  132 S=2*A/D
260  140 GOS 170
265  141 GOS 120
270  142 GOTO 130
275  150 W=(1-M*G/(Z*K))/2
280  151 S=M*V/(Z*K*(W+SQR(W*W+V/Z)))+.005
285  152 GOS 170
290  160 IF I<=0 THEN 130
295  161 GOS 120
300  162 IF J>0 THEN 40
305  163 IF V>0 THEN 150
310  164 GOTO 40
315  170 Q=S*K/M
320  171 J=V+G*S-Z*Q*(1+Q*(.5+Q*(1/3+Q*(.25+Q/5))))
325  180 I=A-G*S*S/2-V*S+Z*S*Q*(.5+Q*(1/6+Q*(1/12+Q/20)))
330  181 RET
     200 END

READY
```